
7
String Representation of DN

7.1
Overview

This clause specifies the string representation of DN. This work is based on IETF RFC 2253 [7]. A DN string representation, using the string-encoding scheme specified in the present document, is also a valid DN string according to IETF RFC 2253 [7].

The string-encoding scheme specified in the present document imposes further restrictions as compared to IETF RFC 2253 [7]. The most important restrictions are:

· Multi-valued RDN is not supported in the subject name convention.

· Character asterisk is used to denote wildcard in the subject name convention.
7.2
Allowed character sets

Subject to further restrictions described in the following subclauses, the allowed characters for the string representation of DN are:

-
Characters of ISO/IEC 646 [14] International Reference Version (IRV) coded character set, and

-
Characters of standard coded character sets supporting and extending ISO/IEC 646 [14] IRV coded character set, e.g. ISO/IEC 10646 [15] coded character set.

NOTE 1:
ISO/IEC 646 [14] IRV coded character set is the international equivalent to the ANSI X3.4 ASCII coded character set.

NOTE 2:
The character set of ISO/IEC 646 [14] IRV corresponds to the subset of characters that range from U+0000 to U+007F in the character set of ISO/IEC 10646 [15].

NOTE 3:
The ISO/IEC 646 [14] IRV characters specifically referenced in this specification are further identified using ISO/IEC 10646 [15] character short identifier notation form "U+XXXX".

7.3
Converting DN from ASN.1 to String

Subclause 7.3.1 defines the algorithm to convert an ASN.1 structured representation to one-string DN representation.  Subclause 7.3.2 defines the algorithm to convert an ASN.1 structured representation to multi-string DN representation.  
CORBA SS uses one-string DN representation.  XML SS uses both one-string and multi-string DN representations.
7.3.1
Rule for one-string DN
7.3.1.1   Converting RDNSequence
If the RDNSequence is an empty sequence, the result is the empty or zero length string.

Otherwise, the output consists of the string encoding of each RDN in the RDNSequence (according to subclause 7.3.1.2), starting with the first element of the sequence and moving forward toward the last element.

The encoding of adjacent RDNs are separated by a comma character (',', U+002C), to be consistent with IETF RFC 2253 [7].
White spaces adjacent to the comma character shall be ignored.

7.3.1.2
Converting RelativeDistinguishedName

When converting from an RDN to a string, the output consists of the string encoding of the singleton AttributeTypeAndValue (according to subclause 7.1, i.e. “Multi-valued RDN is not supported in the subject name convention.” ).
Although ITU-T Recommendation X.500 DN supports multi-valued RDN, this specification supports single-valued RDN only.

7.3.1.3
Converting AttributeTypeAndValue

The AttributeTypeAndValue is encoded as the string representation of the AttributeType, followed by an equals sign character ('=', U+003D), followed by the string representation of the AttributeValue.

Although ITU-T Recommendation X.500 ASN.1 AttributeValue and AttributeType support wide range of character representation, this specification supports a restrictive set of characters according to subclause 7.4.

String representation of AttributeValue allows character escape mechanism such as the use of a reverse solidus character ('\', U+005C) followed by two hexadecimal digits to replace a character in a string. String representation of AttributeType does not allow character escape mechanism.

EXAMPLE:
"CN=Before\0DAfter,O=Test,C=GB". In this example, the reverse solidus character and the two hexadecimal digits form a single byte in the code of the escaped character. The reverse solidus character followed by "0D" indicates a carriage return character. See annex B for a rule for MO designers to avoid ambiguity concerning the AttributeType of a DN string.
7.3.2
Rule for multi-string DN
7.3.2.1
Converting RDNSequence
If the RDNSequence is an empty sequence, there is no equivalent multi-string DN output, i.e. there is no multi-string DN representation for empty RDNSequence sequence.
Otherwise, the output consists of the string encoding of all RDNs in the RDNSequence (according to subclause 7.3.2.2).
One RDN encoding is within an NRM class associated XML element.  A multiple RDNSequence would be converted into multiple RDN encodings, using multiple NRM class associated XML elements.  They are arranged in a hierarchy in the XML document.  The outer NRM class associated XML element represents the first RDN in the RDNSequence.  

7.3.2.2
Converting RelativeDistinguishedName
When converting from an RDN to a string, the output consists of the encoding of the singleton AttributeTypeAndValue (according to subclause 7.1, i.e. “Multi-valued RDN is not supported in the subject name convention.”).
Although ITU-T Recommendation X.500 DN supports multi-valued RDN, this specification supports single-valued RDN only.

7.3.2.3
Converting AttributeTypeAndValue
Within an NRM class XML element, there is an element such as <element name="ManagedElement"> and there is another element <attribute name="id" type="string" use="required"/> which is inherited by using <extension base="xn:NrmClass">.

The AttributeType and AttributeValue are mapped to the former and latter XML elements.  The AttributeType carries the IOC name while the AttributeValue carries the value of the IOC naming-attribute.
Although ITU-T Recommendation X.500 ASN.1 AttributeValue and AttributeType support wide range of character representation, this specification supports a restrictive set of characters according to subclause 7.4.

String representation of AttributeValue allows character escape mechanism (see 7.3.1.3).
7.4
Character syntax

This subclause specifies the character syntax for AttributeType and AttributeValue.

They are:

1.
Any character except:

-
comma character (',', U+002C),

-
equals sign character ('=', U+003D),

-
carriage return character (U+000D),

-
line feed character (U+000A),

-
plus sign character ('+', U+002B),

-
less-than sign character ('<', U+003C),

-
greater-than sign character ('>', U+003E),

-
number sign character ('#', U+0023),

-
semicolon character (';', U+003B),

-
reverse solidus character ('\', U+005C),

-
quotation mark character ('"', U+0022).

2.
The full stop character ('.', U+002E). This character shall be used in the AttributeValue whose AttributeType is "DC". An example is "DC=marketing.CompanyXYZ.com". This full stop character shall not be used in AttributeType.

3.
The asterisk character ('*', U+002A) is reserved to denote wildcard. Wildcard character(s) can appear in AttributeType and AttributeValue. The wildcard character can be used to represent one or more characters.

7.5
EBNF of DN String Representation

7.5
EBNF of DN String Representation

The formal definitions provided within this subclause consolidate several rules and concepts (null distinguished name, DN prefix, local DN, domain component type, class names starting with upper case characters, attribute names starting with lower case characters, classes with or without an "Id" naming attribute, attribute type and attribute value allowed characters, wildcard character). The definition is more detailed to clarify these naming rules, and will not introduce compliancy issues for implementations compliant with Rel-5 version of this specification.

The following is the EBNF for DN in string representation (Extended Backus-Naur Form; see ISO/IEC 14977 [13] for more information):

DistinguishedName              =   NullDN       (* Distinguished Names shall not exceed   *)
                                 | RegularDN ;  (* 400 octets as specified in section 7.6 *)

NullDN                         = ;  (* empty string; null DN is specified in subclause 7.3.1 *)

RegularDN                      =   DNPrefixPlusRDNSeparator  (* DN prefix and local DN *)
                                 , LocalDN ;                 (* are defined in annex C *)

DNPrefixPlusRDNSeparator       =   ( NullDNPrefix , NullRDNSeparator )
                                 | ( DNPrefixWithDomainComponent , RDNSeparator )
                                 | ( DNPrefixWithoutDomainComponent , RDNSeparator ) ;

NullDNPrefix                   = ;  (* empty string *)

NullRDNSeparator               = ;  (* empty string *)

DNPrefixWithDomainComponent    =   DomainComponentRDN
                                 , { RDNSeparator , DomainComponentRDN }
                                 , { RDNSeparator , RegularRDN } ;

DNPrefixWithoutDomainComponent =   RegularRDN
                                 , { RDNSeparator , RegularRDN } ;

LocalDN                        =   LocalRDN
                                 , { RDNSeparator , LocalRDN } ;

RDNSeparator                   =   [ RDNSeparatorWhiteSpace ]    (* use of optional white space  *)
                                 , CommaChar                     (* is recommended to be avoided *)
                                 , [ RDNSeparatorWhiteSpace ] ;

RDNSeparatorWhiteSpace         =   [ CarriageReturnChar ]
                                 , { SpaceChar } ;

DomainComponentRDN             = DCAttributeTypeAndValue ;

RegularRDN                     = RegularAttributeTypeAndValue ;

LocalRDN                       = LocalDNAttributeTypeAndValue ;

DCAttributeTypeAndValue        =   DCAttributeType
                                 , AttributeTypeAndValueSeparator
                                 , ( DCAttributeValue | WildcardDCAttributeValue ) ;

RegularAttributeTypeAndValue   =   ( RegularAttributeType | WildcardRegularAttributeType )
                                 , AttributeTypeAndValueSeparator
                                 , ( RegularAttributeValue | WildcardRegularAttributeValue ) ;

LocalDNAttributeTypeAndValue   =   ( LocalDNAttributeType | WildcardLocalDNAttributeType )
                                 , AttributeTypeAndValueSeparator
                                 , ( RegularAttributeValue | WildcardRegularAttributeValue ) ;

AttributeTypeAndValueSeparator = EqualsSignChar ;

DCAttributeType                = "DC" ;  (* ISO/IEC 646 IRV U+0044/0043 Latin capital letters D&C *)

DCAttributeValue               =   DCLabel                           (* this is specified *)
                                 , { DCLabelSeparator , DCLabel } ;  (* in IETF RFC 1035  *)

WildcardDCAttributeValue       =   (   ( DCLabel | WildcardDCLabel )
                                     , { DCLabelSeparator , ( DCLabel | WildcardDCLabel ) } )
                                 - DCAttributeValue ;

DCLabelSeparator               = FullStopChar ;  (* this is specified in IETF RFC 1035 *)

DCLabel                        =   LetterChar                             (* this is specified *)
                                 , [   { LetterDigitHypenMinusChar }      (* in IETF RFC 1035  *)
                                     , LetterDigitChar               ] ;

WildcardDCLabel                =   (   ( LetterChar | WildcardChar )
                                     , [   { LetterDigitHypenMinusChar | WildcardChar }
                                         , ( LetterDigitChar | WildcardChar )           ] )
                                 - DCLabel ;

RegularAttributeType           =   LetterChar                       (* this is specified *)
                                 , { LetterDigitHypenMinusChar } ;  (* in IETF RFC 2253  *)

WildcardRegularAttributeType   =   (   ( LetterChar | WildcardChar )
                                     , { LetterDigitHypenMinusChar | WildcardChar } )
                                 - RegularAttributeType ;

LocalDNAttributeType           =   NameOfClassWithIdAttribute        (* definition selected shall *)
                                 | NamesOfClassAndNamingAttribute ;  (* be in accordance with the *)
                                                                     (* rules defined in annex B  *)

WildcardLocalDNAttributeType   =   WildcardNameOfClassWithIdAttr
                                 | WildcardNamesOfClassAndNamAttr ;

NameOfClassWithIdAttribute     = ClassName ;  (* see rules defined in annex B *)

WildcardNameOfClassWithIdAttr  = WildcardClassName ;

NamesOfClassAndNamingAttribute =   ClassName                      (* see rules defined in annex B *)
                                 , ClassNamingAttributeSeparator
                                 , NamingAttributeName ;

WildcardNamesOfClassAndNamAttr =   (   ( ClassName | WildcardClassName )
                                     , ClassNamingAttributeSeparator
                                     , ( NamingAttributeName | WildcardNamingAttributeName ) )
                                 - NamesOfClassAndNamingAttribute ;

ClassNamingAttributeSeparator  = FullStopChar ;  (* see rules defined in annex B *)

ClassName                      =   CapitalLetterChar               (* see recommendation on       *)
                                 , { LocalDNAttributeTypeChar } ;  (* characters for class names  *)
                                                                   (* in annex E   

WildcardClassName              =   (   ( CapitalLetterChar | WildcardChar )
                                     , { LocalDNAttributeTypeChar | WildcardChar } )
                                 - ClassName ;

NamingAttributeName            =   SmallLetterChar
                                 , { LocalDNAttributeTypeChar } ;

WildcardNamingAttributeName    =   (   ( SmallLetterChar | WildcardChar )
                                     , { LocalDNAttributeTypeChar | WildcardChar } )
                                 - NamingAttributeName ;

RegularAttributeValue          =   ( AttributeValueChar – SpaceChar )          (* this is       *)
                                 , [   { AttributeValueChar }                  (* specified in  *)
                                     , ( AttributeValueChar – SpaceChar ) ] ;  (* IETF RFC 2253 *)

WildcardRegularAttributeValue  =   (   ( ( AttributeValueChar – SpaceChar ) | WildcardChar )
                                     , [   { AttributeValueChar | WildcardChar }
                                         , ( ( AttributeValueChar – SpaceChar ) | WildcardChar ) ] )
                                 - RegularAttributeValue ;

LocalDNAttributeTypeChar       = DNChar – FullStopChar ;

AttributeValueChar             = DNChar | EscapedCharSequence ;

WildcardChar                   = AsteriskChar ;  (* this is specified in subclause 7.4 *)

DNChar                         = DNCharUnrestricted - ReservedChar ;

DNCharUnrestricted             =   ? Character of ISO/IEC 646 IRV ?
                                 | ? Character of standard coded character set
                                     supporting and extending ISO/IEC 646 IRV  ? ;

EscapedCharSequence            =   ReverseSolidusChar          (* this is specified  *)
                                 , 2 * HexadecimalDigitChar ;  (* in subclause 7.3.3 *)

ReservedChar                   =   Rfc2253ReservedChar | CarriageReturnChar | LineFeedChar
                                 | AsteriskChar ;

Rfc2253ReservedChar            =   CommaChar | EqualsSignChar | PlusSignChar | LessThanSignChar
                                 | GreaterThanSignChar | NumberSignChar | SemiColonChar
                                 | ReverseSolidusChar | QuotationMarkChar ;

LetterChar                     = CapitalLetterChar | SmallLetterChar ;

LetterDigitChar                = LetterChar | DigitChar ;

LetterDigitHypenMinusChar      = LetterDigitChar | HypenMinusChar ;

HexadecimalDigitChar           = DigitChar | CapitalLetterAtoFChar | SmallLetterAtoFChar ;

LineFeedChar                   = ? ISO/IEC 646 IRV U+000A character line feed ? ;

CarriageReturnChar             = ? ISO/IEC 646 IRV U+000D character carriage return ? ;

SpaceChar                      = ' ' ;  (* ISO/IEC 646 IRV U+0020 character space *)

QuotationMarkChar              = '"' ;  (* ISO/IEC 646 IRV U+0022 character quotation mark *)

NumberSignChar                 = '#' ;  (* ISO/IEC 646 IRV U+0023 character number sign *)

AsteriskChar                   = '*' ;  (* ISO/IEC 646 IRV U+002A character asterisk *)

PlusSignChar                   = '+' ;  (* ISO/IEC 646 IRV U+002B character plus sign *)

CommaChar                      = ',' ;  (* ISO/IEC 646 IRV U+002C character comma *)

HypenMinusChar                 = '-' ;  (* ISO/IEC 646 IRV U+002D character hyphen-minus *)

FullStopChar                   = '.' ;  (* ISO/IEC 646 IRV U+002E character full stop *)

DigitChar                      =   '0' | '1' | '2' | '3' | '4'    (* ISO/IEC 646 IRV U+0030-0039 *)
                                 | '5' | '6' | '7' | '8' | '9' ;  (* digits 0 to 9               *)

SemiColonChar                  = ';' ;  (* ISO/IEC 646 IRV U+003B character semicolon *)

LessThanSignChar               = '<' ;  (* ISO/IEC 646 IRV U+003C character less-than sign *)

EqualsSignChar                 = '=' ;  (* ISO/IEC 646 IRV U+003D character equals sign *)

GreaterThanSignChar            = '>' ;  (* ISO/IEC 646 IRV U+003E character greater-than sign *)

CapitalLetterAtoFChar          =   'A' | 'B' | 'C'    (* ISO/IEC 646 IRV U+0041-0046  *)
                                 | 'D' | 'E' | 'F' ;  (* Latin capital letters A to F *)

CapitalLetterChar              =   CapitalLetterAtoFChar | 'G' | 'H'    (* ISO/IEC 646 IRV *)
                                 | 'I' | 'J' | 'K' | 'L' | 'M' | 'N'    (* U+0041-005A     *)
                                 | 'O' | 'P' | 'Q' | 'R' | 'S' | 'T'    (* Latin capital   *)
                                 | 'U' | 'V' | 'W' | 'X' | 'Y' | 'Z' ;  (* letters A to Z  *)

ReverseSolidusChar             = '\' ;  (* ISO/IEC 646 IRV U+005C character reverse solidus *)

SmallLetterAtoFChar            =   'a' | 'b' | 'c'    (* ISO/IEC 646 IRV U+0061-0066 *)
                                 | 'd' | 'e' | 'f' ;  (* Latin small letters a to f  *)

SmallLetterChar                =   SmallLetterAtoFChar | 'g' | 'h'      (* ISO/IEC 646 IRV *)
                                 | 'i' | 'j' | 'k' | 'l' | 'm' | 'n'    (* U+0061-007A     *)
                                 | 'o' | 'p' | 'q' | 'r' | 's' | 't'    (* Latin small     *)
                                 | 'u' | 'v' | 'w' | 'x' | 'y' | 'z' ;  (* letters a to z  *)

7.6
Maximum size of DN string

The maximum length of a DN string, including RDN separators and including white spaces, shall not exceed 400 bytes (8-bit).

Annex B (normative):
Rule for MO Designers regarding AttributeType interpretation

NOTE:
This annex is normative for users of one string representation.
This annex discusses the two possible interpretations for the AttributeType of the DN string and establishes a rule for MO designers to avoid ambiguity concerning their  usage. 
It also gives a rule for designing MO classes such that one DN string, regardless of the IRP SS technology used, will result in a  unique reference to the corresponding network resource.

First interpretation
ITU-T Recommendation X.500 [2] uses the AttributeType (defined for use as the first component of the AttributeTypeAndValue of a RDN, see subclause 3.1.6) to identify one attribute of the subject MO for naming purpose. This AttributeType is called the naming attribute to distinguish itself from other attributes that may be present in the MO.

Suppose the following is the MO class definition in pseudo notation and this MO class is inherited from root.

Class Bsc {


Attribute id;


Attribute ..}

Suppose further that the naming attribute is id.

If this (first) interpretation is used for constructing the DN string, then the DN will be "…,id=123". MO class name cannot be derived from the DN string. The value of the AttributeValue contains the value of the naming attribute.

Second interpretation

Generally, this interpretation shall be preferred to the first interpretation.

The AttributeType (defined for use as the first component of the AttributeTypeAndValue of a RDN) is used to identify the MO class.Using this interpretation for constructing the DN string  the DN will be "…,Bsc=123". 
The name of the naming attribute cannot be derived from the DN string. 
The value of the AttributeValue contains the value of the naming attribute.

Rule

Given the two interpretations, a DN reader cannot know how to interpret the AttributeType, i.e. if the AttributeType identifies class or naming attribute. To avoid ambiguity, the following rules shall apply: 
· If the IS name of the IOC naming attribute is "id", ignoring case of “id”, then the DN shall use "…,Yyy=123,…" where "Yyy" is the IS name of the IOC, preserving case of “Yyy”.

EXAMPLE 1:
If "Bsc" is the IS name of the IOC and if the IS name of its naming attribute is "id” then the DN shall be "…,Bsc=123,…".

· If the IS name of the IOC naming attribute is not the concatenation of the IS name of the IOC and "Id", ignoring case for both, and if the IS name of the IOC naming attribute is not  "id", ignoring case of “id”, then the DN shall use "…,Yyy.zzz=123,…" where "Yyy" is the IS name of the IOC and "zzz" is the IS name of the IOC naming attribute, preserving case for both.

EXAMPLE 2:
If "Bsc" is the IS name of the IOC and if the IS name of its naming attribute is "serialNumber", then the DN shall be "…,Bsc.serialNumber=123,…".

· If the IS name of the IOC naming attribute is the concatenation of the IS name of the IOC and "Id", ignoring case for both, then the DN shall use "…,Xxx=123,…" where "Xxx" is the IS name of the IOC, preserving case.

EXAMPLE 3:
If "Bsc" is the IS name of the IOC and if the IS name of its naming attribute is "bscId", then the DN shall be "…,Bsc=123,…".

